
Introduction


In reading about other people’s work on simulating real-world physics such as [1], I found a number of references to numerical solvers for differential equations. It seems that you can’t simulate springs, bounces, collisions, deformations, etc. if you don’t have a basic understanding of calculus and differential equations (my apologies to those who don’t). Before I can simulate a flag blowing in the wind or falling off a flagpole, I need a solver. The most robust and most used numerical solver for that purpose is a fourth order Runge-Kutta algorithm that numerically integrates using a step size h with an error that is on the order of h4. My numerical analysis teacher, who was actually a numerical analyst, pronounced the “ge” in “Runge” as in “get” without the “t” so that the name actually came out sounding like Runga-Kutta.


<PULL QUOTE from above: “The most robust and most used numerical solver is a fourth order Runge-Kutta algorithm.”>


I thought I would implement a variation of this algorithm that dynamically adjusts its step size and that can be used on many equations to be solved at once, i.e., on a vector of differential equations. 


Differential Equations: What’s that?


An equation that involves time is a differential equation. So even something as mundane as “F = ma” is a differential equation because “a” is really “dv/dt” which is sometimes written with the quoted notation; i.e., v'. Similarly, an equation involving v by itself is a differential equation because v is really dx/dt; i.e., x'. For simple equations involving x, v, and a, we can solve the equation using high-school mathematics and avoid the issue of differential equations entirely. But as soon as we add a simple complication like making the acceleration depend on the velocity, we can no longer solve the equation. For example, in a spring, the force is proportional to the velocity of the spring's expansion. So we might have something like ma = -cv for some known constant c which essentially says that the acceleration decreases more for high velocities than low ones. If we write this simply as a = -kv by dividing through by m (and letting k = c/m), we find that we have a differential equation of the form


v' = -kv


You can't solve this with high-school mathematics. The solution is in fact


v = Ke-kt


where K is another constant (this time, unknown) and e is the natural logarithm 2.718… This turns out to be a simple differential equation to solve. But many differential equations are just too difficult to solve exactly. If we can at least write down the equations, it would be nice to have a program that can solve it. That's where the Runge-Kutta algorithm comes in.


<KINDOF A PULL QUOTE from above “You can't solve differential equations with high-school mathematics.”>


In general, a differential equation can be written in the form


y' = f (t, y)


where y = y0 when t = t0 (the initial conditions). The numerical solver can be told to solve for y at times t1, then t2, then t3, … to get successive solutions y1, y2, y3, … over time. 


Even when the solution is known, the initial conditions are needed to compute the unknown constant. In the above example, if v = v0 at t = t0, we can compute the constant K as v0/e-kt0.


The Runge-Kutta Algorithm


All of the algorithms that I've seen come from code extracted from the samples in the well-known book titled “Numerical Recipes in C” [2]. Additionally, all the articles start with the preamble “it can be shown that” without adding “but we don’t have a clue how!” because that's essentially what reference [2] contains. I found another book that fully derives the order 1 formula (accurate to h), indicates that the order 2 formula can be derived in a similar way but doesn’t do it, and then confesses that order's 3 and 4 are too complicated to do. Several solutions are, however, given. It's all a bit of a mess. Forgive me for imitating everyone else but given a solution y i at time t i  (initially, we start with y0 at time t 0), the solution at time t i  + h is 


yi+1 = yi  + k1/6 + k2/3 +k3/3 + k4/6 + O(h5) at


t i+1 = t i + h





where


k1 = h f (t i,y i) 


k2 = h f (t i+h/2, y i+ k1/2)


k3 = h f (t i+h/2, y i+ k2/2) 


k4 = h f (t i+h, y i+ k3)





Notice that f(t,y) is the derivative y' (in the example above, it’s –kv). Also, note that k2 is in terms of k1, k3 is in terms of k2, and k4 in terms of k3. If you care, the derivation proceeds by assuming a solution as shown with unknowns replacing the 4 constants in yi+1, and two constants in each of k2, k3, and k4 (k1 is kept unchanged). Then yi+1 = y i (t i + h) is expanded in a Taylor series (engineers know what this means). At the same time, the ki’s are expanded in Taylor series and substituted into the yi+1 formula above. By forcing the coefficients to match up in the two Taylor series, we end up with 9 equations in 10 unknowns. One unknown can be set up arbitrarily and the others must be solved. I’ve seen 3 very different solutions for the constants and the above is the simplest.


To apply such a solver to simulating a piece of cloth falling to the ground, we might model the cloth as a 10 by 10 sheet of nodes, where each node is connected to a neighbor by a small spring. We would need to solve 100 such equations (one per node) in parallel. We can do that by generalizing the above to vectors of equations. So yi and yi+1 become vectors. Since we need a different k1 for each yi, k1 becomes a vector too, and so does k2, and k3, and k4. In summary, we use vectors for yi+1, yi, k1, k2, k3, and k4.


Support for the Implementation


Because numerical algorithms are slow, in general, it’s important to design the code in such a way that we avoid dynamically constructing collections. So, we’ll try to avoid code like





	vector1 := vector2 collect: [:item | 


		… code using item …]





However, the code looks so clumsy if we actually write directly into our vectors as follows:





	1 to: vector1 size do: [:index |


		vector1


			at: index


			put: (… code using 


				(vector2 at: index) …)





So we created collection methods that ease the task. A few examples are shown below.


	vector1 assign: vector2.


	vector1 set: [0]. 


	vector1 with: vector2 set: [:x | x + 1]


	vector1 with: vector2 with: vector3 set: [:x :y | x * y + 1]








The last method, for example, allows us to set element “i” of vector1 to the product of the corresponding elements in vector2 and vector3 with 1 added on. We basically specify how one element is computed from the parameters, which are elements from the “with:” collections. The implementation worries about iterating, indexing, and storing the result in the receiver. These methods and a few others are shown in Listing 1.


The Implementation


Listing 2 contains an implementation of our differential equation solver, which we’ll get to in a moment. To use it, we first obtain an instance (a solver), for example, as follows:








	^DifferentialEquationSolver 


		solverFor: [:dydt :t :y | 


			"Computes y' and solves for y in y' = f(t,y)..."


			"a = v' = -av where a = 2 whose solution is v = k1 e-at."


			dydt at: 1 put: a negated * y first.


			"v = x' = b where b = 3 whose solution is x = bt + k2."


			dydt at: 2 put: b]





We provide the block to use for computing vector y'; i.e., dydt, from the value t and y (also, a vector). This particular example is a 2-equation solver. The first equation to be solved is v' = -av. So –av is stored in the first element of dydt. The second equation to be solved is x' = b. So b is stored in the second element. So this solver will solve for v and x; i.e., vector y represents [v,x]. 





How does the solver know that we are solving 2 simultaneous equations as opposed to 10? It doesn’t. Not until we ask it to compute a value, which we do as follows:





	aSolver


		“solveAt: t given: y0 and: t0”


		solveAt: 1.0 given: #(10.0 0.0) at: 0.0





The initial value for y0 is used to determine the vector size, which explains how the solver is able to provide a 2-vector for dydt in the block mentioned above. The solver actually pre-allocates all temporary vectors needed when the solve message is sent. After an answer y is returned, we can be feed this y with 1.0 back into the solver as initial values for the next solution to obtain a solution at a later time, e.g., at time 2.0. No new allocations will occur the second time the solve message is sent. 





< PULL QUOTE from above “The solver actually pre-allocates all temporary vectors needed when the solve message is sent.”>





If we now consider the class definition in Listing 2, we will see that there are a fair number of instance variables. Except for the instance variables that deal with blocks, h, accuracy, and those representing boolean values, all others represent vectors that are pre-allocated in method “reinitializeFor: sampleSolution” when the vector size changes.





Another question that might come to mind if you’ve dealt with numerical techniques before is “how is the step size determined since that affects the accuracy of the result?” Don’t we have to supply a specific value for h? In our case, the answer is no. The step size starts at an arbitrary value, like 1.0, but it is monitored and dynamically adjusted until the requested relative accuracy is achieved (or it is determined that it is not achievable because h is impossibly small). This relative accuracy requirement can be changed at any time; it is initially set at 0.0001; i.e., a 0.01 per cent error is permitted. 





As a rule, we decrease h only when our error tolerance is not met. Once we find an h value that works, we leave it alone until the solver has successful met the tolerance “n” consecutive times in a row. We call “n” by a somewhat longer name; namely, “successesBeforeIncreasingH.” We use halving and doubling (roughly) to decrease and increase h so as to provide a logarithmic zeroing in to a suitable h value.





Method “solveAt: t given: v0 and: t0” decides if it is time to increase h (which it deals with appropriately) and also decides whether the t value requested has increased or decreased relative to t0. It then picks either the method that deals with an increasing t or one that deals with a decreasing t. Using two methods is more readable than using a single combined version that multiplies every test by a sign as done in the Numerical Recipes book. 





Since both methods are similar, consider “solveAtIncreasing: t given: v0 and: t0.” Starting with t0, the method repeatedly adds h until t1 is reached while at the same time computing a new y at each step (the new y replaces the old). One complication is that h need not be constant. Although the real work of computing the new y is done by “step: y at: t”, this method can decrease h if it feels that is it necessary to meet the accuracy requirements. That’s why h is added after the new y has been computed (rather than before). 





Method “step: y at: t” deals with monitoring the h value for accuracy. The way it does this is to compute two solutions using method “into: solution rungaKuttaStep: y and: dydt at: t h: h.” The first solution steps by h/2 twice while the second steps by h once. Both should lead to the same result if the computations were exact.  To find out, we compare the respective answers in “solution1” and “solution2.” If the maximum difference is less than the tolerated error, we replace y by the new solution and return. Usually, the solution computed with the smallest h will be more accurate unless the h used is the smallest allowed. In that case, the solution with the largest h is best. As it turns out, a fifth order Runge-Kutta solution can be obtained by using both values to compute an answer rather than using either one individually, which is what we do.





A few other interesting details can be observed in the method. First, every time the method fails to meet the error condition, it decreases the h and tries again unless h is already too small. Second, the method illustrates the first encounter of message “dydtAt: t and: y andSet: result” which can be paraphrased as “compute dydt at the specified t and y values and place the answer in vector result.” If we were to trace the message, we would see that it ends up executing the block that we provided when we first obtained an instance of the solver.





As might be expected, the code that actually implements the Runge-Kutta algorithm is contained in method “into: solution rungaKuttaStep: y and: dydt at: t h: h.” The 22 line method (with 5 comment lines) is rather dense but nevertheless a straight translation of “yi+1 = yi  + k1/6 + k2/3 +k3/3 + k4/6” with the direct computation of each of the k i vector values.





How Well Does it Work?


We used the code “DifferentialEquationSolver example: 2 solveFrom: 0.0 to: 10.0 by: 1.0 initially: #(10.0 10.0)” to test our implementation as we evolved it. It prints both the solution provided by the solver and the exact solution that should result. Initially, the solver’s solution was quite different from the exact solution and we spent a couple of days trying to track the source of the bug while at the same time improving how we monitored and modified h values. But it turned out that our exact solution was wrong. Once we fixed it, our generated output proved to be virtually identical. By modifying our terseString method to print more digits (as coded in Listing 2, only a few digits of accuracy would have shown), Table 1 results.





Time: Exact 			=> Computed


 0.0: [10.0, 10.0] 		=> [10.0, 10.0].


 1.0: [1.35335283, 13.0] 	=> [1.35335211, 13.0].


 2.0: [0.18315639, 16.0] 	=> [0.18315619, 16.0].


 3.0: [0.24787522e-1, 19.0] 	=> [0.24787482e-1, 19.0].


 4.0: [0.33546263e-2, 22.0] 	=> [0.33546191e-2, 22.0].


 5.0: [0.4539993e-3, 25.0] 	=> [0.45399816e-3, 25.0].


 6.0: [0.61442124e-4, 28.0] 	=> [0.61441945e-4, 28.0].


 7.0: [0.83152872e-5, 31.0] 	=> [0.83152598e-5, 31.0].


 8.0: [0.11253517e-5, 34.0] 	=> [0.11253476e-5, 34.0].


 9.0: [0.1522998e-6, 37.0] 	=> [0.15229919e-6, 37.0].


10.0: [0.20611536e-7, 40.0] 	=> [0.20611448e-7, 40.0].


Table 1 Comparing approximate and exact solutions.


Conclusions


By implementing a relatively flexible differential equation solver for vectors of differential equations based on a fourth order Runge-Kunta algorithm with the capability to monitor and adaptively modify h using a logarithmic based zeroing-in capability, we’ve provided the tools needed for solving interesting physics problems. Now, if we only had an interesting physics problem.

Listing 1 Extensions





Object extensions


terseString


	^self printString





Float extensions


terseString


	| string point exponent |


	string := self printString.


	point := string indexOf: $..


	exponent := string indexOf: $e.


	exponent = 0 ifTrue: [^string copyFrom: 1 to: point + 1].


	^(string copyFrom: 1 to: point + 1), 


		(string copyFrom: exponent to: string size)





Collection extensions


allButFirstDo: aBlock


	self isEmpty ifTrue: [^self].


	(self copyFrom: 2 to: self size) do: aBlock





IndexedCollectionr extensions


assign: array


	"Equivalent to 'self with: array1 set: [:item | item]'."


	1 to: self size do: [:index |


		self at: index put: (array at: index)]





set: aBlock


	"anArray set: [20]"


	1 to: self size do: [:index |


		self at: index put: (aBlock value)]


   


with: array1 set: aBlock


	"Instead of writing


		For each i,     test1 at: i put: (test2 at: i) computeValue.


	User writes


		test1 with: test2 set: [:value2 | value2 computeValue]


	Method takes care of iterating, indexing, and storing 


	the value in test1. Multiple with:'s provide 


	corresponding multiple values to work from."





	1 to: self size do: [:index |


		self


			at: index


			put: (aBlock value: (array1 at: index))]





other variations


with: array1 with: array2 set: aBlock


with: array1 with: array2 with: array3 set: aBlock


…


	…code similar to above…





Listing 2 Class DifferentialEquationSolver.





class:								DifferentialEquationSolver


superclass:						Object 


instance variables:		differentiationBlock successes 											successesBeforeIncreasingH solution0 


										solution1 solution2 k1 k2 k3 k4 y y1 y2 


										y3 dydt dydt0 dydt1 dydt2 dydt3 h 


										minimumH relativeAccuracy





class methods





instance creation


new


	^super new initialize





solverFor: aBlock


	^self new differentiationBlock: aBlock





examples


example1


	"DifferentialEquationSolver example1


		solveAt: 1.0 given: #(10.0 0.0) at: 0.0"


	| a b |


	a := 2.0. b := 3.0.


	^self solverFor: [:dydt :t :y |


		"Computes y' and solves for y in y' = f(t,y)..."


		"a = v' = -av where a = 2 whose solution is v = k1 e-at."


		dydt at: 1 put: a negated * y first.


		"v = x' = b where b = 3 whose solution is x = bt + k2."


		dydt at: 2 put: b]





example: number solveFrom: t0 to: t1 by: step initially: y0


	"DifferentialEquationSolver example: 1 


		solveFrom: 0.0 to: 10.0 by: 1.0 initially: #(0.0).


	DifferentialEquationSolver example: 2 


		solveFrom: 0.0 to: 10.0 by: 1.0 initially: #(10.0 10.0)"


	| which solverName exactSolutionMethod solver t y solution1


	solution2 |


	which := number printString.


	solverName := ('solver', which) asSymbol.


	exactSolutionMethod := ('exactSolution', which,


		'At:given:at:') asSymbol.





	self printTitle: 'Solving example ', which.


	solver := self perform: solverName.





	t := t0. y := y0.


	self printTime: t solutions: y and: y.





	"The exact solution is computed from y0 and t0 whereas


	the approximate solution is computed from the last y."


	t0 + step to: t1 by: step do: [:newT |


		solution1 := self perform: exactSolutionMethod 


			with: newT with: y0 with: t0.


		solution2 := solver solveAt: newT given: y at: t.


		self printTime: newT solutions: solution1 and: solution2.


		t := newT. y := solution2].


	Transcript cr; show: 'All done'





printing


printTime: t solutions: y1 and: y2


	Transcript cr; show: t terseString; show: ': '.


	Transcript show: '['; show: y1 first terseString.


	y1 allButFirstDo: [:number |


		Transcript show: ', '; show: number terseString].


	Transcript show: '] => ['; show: y2 first terseString.


	y2 allButFirstDo: [:number |


		Transcript show: ', '; show: number terseString].


	Transcript show: '].'





printTitle: aString


	Transcript


		cr; cr; show: aString;


		cr; show: 'Time: Exact => Computed'.





solutions


solver1


	"DifferentialEquationSolver solver1 


		solveAt: 1.0 given: #(0.0) at: 0.0"


	"Computes y'. Used elsewhere to solve for y in y' = f(t,y)..."


	^self solverFor: [:dydt :t :y |


		"x' = 1 whose solution is x = t + k. With initial conditions


		t=0, x = 0, k = x - t = 0."


		dydt at: 1 put: 1.0]





solver2


	"DifferentialEquationSolver solver2


		solveAt: 1.0 given: #(10.0 0.0) at: 0.0"


	"Computes y'. Used elsewhere to solve for y in y' = f(t,y)..."


	| a b |


	a := 2.0. b := 3.0.


	^self solverFor: [:dydt :t :y |


		"Computes y' and solves for y in y' = f(t,y)..."


		"a = v' = -av where a = 2 whose solution is v = k1 e-at."


		dydt at: 1 put: a negated * y first.


		"v = x' = b where b = 3 whose solution is x = bt + k2."


		dydt at: 2 put: b]





exactSolution1At: t given: y0 at: t0


	"Solve for vector y at time t given solution vector y0 at time


	t0 where y is the differential equation of the form y' = f (t, y).





	Let y be vector <x> where problem is


	x' = 1 whose solution is x = t + k. With initial conditions


	t=t0, x = x0,  k = x0 - t0."





	| k |


	k := y0 first - t0.


	^Array with: t + k





exactSolution2At: t given: y0 at: t0


	"Solve for vector y at time t given solution vector y0 at time


	t0 where y is the differential equation of the form y' = f (t, y).





	Let y be vector <v, x> such that


	v' = -av where a = 2,   x' = b where b = 3


	Solution is v = k1e-at ,  x = bt + k2"


	| a b k1 k2 |


	a := 2.0. b := 3.0.


	k1 := y0 first / (a negated * t0) exp.


	k2 := y0 last - (b * t0).


	^Array


		with: k1 * (a negated * t) exp


		with: (b * t) + k2





instance methods





initializing/finalizing


initialize


	self h: 1.0.


	self minimumH: 1.0e-5.


	self successesBeforeIncreasingH: 3.


	successes := 0.


	self relativeAccuracy: 0.0001.


	self reinitializefor: #()





reinitializefor: sampleSolution


	y size = sampleSolution size ifTrue: [^self].


	solution0 := sampleSolution shallowCopy.


	solution1 := sampleSolution shallowCopy.


	solution2 := sampleSolution shallowCopy.


	y := sampleSolution shallowCopy.


	k1 := sampleSolution shallowCopy.


	k2 := sampleSolution shallowCopy.


	k3 := sampleSolution shallowCopy.


	k4 := sampleSolution shallowCopy.


	y1 := sampleSolution shallowCopy.


	y2 := sampleSolution shallowCopy.


	y3 := sampleSolution shallowCopy.


	dydt := sampleSolution shallowCopy.


	dydt0 := sampleSolution shallowCopy.


	dydt1 := sampleSolution shallowCopy.


	dydt2 := sampleSolution shallowCopy.


	dydt3 := sampleSolution shallowCopy.





get/set


h


	^h


h: aNumber


	h := aNumber


minimumH: aNumber


	minimumH := aNumber


minimumH


	^minimumH


relativeAccuracy


	^relativeAccuracy


relativeAccuracy: aNumber


	relativeAccuracy := aNumber


successesBeforeIncreasingH


	^successesBeforeIncreasingH


successesBeforeIncreasingH: anInteger


	successesBeforeIncreasingH := anInteger


differentiationBlock


	^differentiationBlock


differentiationBlock: aBlock


	differentiationBlock := aBlock





evaluating


dydtAt: t and: y andSet: result


	self differentiationBlock value: result value: t value: y





solving - public


solveAt: t given: y0 at: t0


	self reinitializefor: y0.


	successes := successes + 1.


	successes > self successesBeforeIncreasingH ifTrue: [


		"With enough successes in a row, we can increase h."


		successes := 0.


		h := 1.9 *h. "a little less than double"].


	t > t0


		ifTrue: [^self solveAtIncreasing: t given: y0 at: t0]


		ifFalse: [^self solveAtDecreasing: t given: y0 at: t0]





solving - private step 1


solveAtIncreasing: t1 given: y0 at: t0


	| t restoreHIfDone oldH |


	restoreHIfDone := false. t := t0. y assign: y0.


	[true] whileTrue: [


		"Avoid overshoot by adjusting h (restore before leaving)."


		t + h > t1 ifTrue: [


			restoreHIfDone := true. oldH := h. h := t1 - t].


		"Next step can decrease h (which means we will loop)."


		self step: y at: t.


		t := t + h.


		"Quit if done but restore h if saved."


		t  >= t1 ifTrue: [restoreHIfDone ifTrue: [h := oldH]. ^y].


		restoreHIfDone := false]





solveAtDecreasing: t1 given: y0 at: t0


	| t restoreHIfDone oldH |


	restoreHIfDone := false. t := t0. y assign: y0.


	[true] whileTrue: [


		"Avoid overshoot by adjusting h (restore before leaving)."


		t - h < t1 ifTrue: [


			restoreHIfDone := true. oldH := h. h := t - t1].


		"Next step can decrease h (which means we will loop)."


		self step: y at: t.


		t := t - h.


		"Quit if done but restore h if saved."


		t  <= t1 ifTrue: [restoreHIfDone ifTrue: [h := oldH]. ^y].


		restoreHIfDone := false]








solving - private step 2


step: y at: t


	"Given a solution y at t, compute dydt, and find an h 


	beginning with current value or smaller (a minimum exists) 


	for which a solution at t + h is accurate to our 


	specification. Approach is to solve twice: (1) once in two 


	half steps: t0=t+h/2, then t1 = t0+h/2, (2) once in a full 


	step, t2 = t+h. If the t1 and t2 results are close 


	enough, combine them for a fifth order Runga-Kutta 


	solution. Otherwise, decrease the h and try again. Notice 


	that the t+h value for the solution will depend on the value 


	of h when done."


	| halfH t0 error |


	self dydtAt: t and: y andSet: dydt.


	[true] whileTrue: [


		"Take 2 half steps..."


		halfH := 0.5 * h. t0 := t + halfH.


		self into: solution0 


			rungaKuttaStep: y and: dydt at: t h: halfH.


		self dydtAt: t0 and: solution0 andSet: dydt0.


		self into: solution1 


			rungaKuttaStep: solution0 and: dydt0 at: t0 h: halfH.


		"Take 1 full step..."


		self into: solution2 rungaKuttaStep: y and: dydt at: t h: h.


		"Compare the two solutions."


		(halfH < self minimumH or: [self errorIsSmall]) ifTrue: [


			y with: solution1 with: solution2 set: [:y1 :y2 |


				"Solution is Runga-Kutta fifth order."


				y1 + ((y1 - y2) / 15.0)].


			^self].


		"No, determine next value of h to try."


		h := halfH. successes := 0]





computeMaximumRelativeError


	| maximumRelativeError absoluteError absoluteValue 


	relativeError |


	maximumRelativeError := 0.0.


	solution1 with: solution2 do: [:y1 :y2 |


		absoluteError := (y1 - y2) abs.


		"Near 0, use absolute error; otherwise, relative error."


		absoluteValue := (y1 abs max: y2 abs).


		relativeError := absoluteValue < 1.0e-30


			ifTrue: [absoluteError]


			ifFalse: [absoluteError / absoluteValue].


		maximumRelativeError := 


			maximumRelativeError max: relativeError].


	^maximumRelativeError





errorIsSmall


	^self computeMaximumRelativeError 


		< self relativeAccuracy





solving - private step 3


into: solution rungaKuttaStep: y0 and: dydt0 at: t0 h: h


	| t1 t2 t3 |


	t1 := t0 + (0.5*h). t2 := t1. t3 := t0 + h.


	"k1 = h*f'(t0,y0)"


	k1 with: dydt0 set: [:_dydt0 | h * _dydt0].


	"k2 = h*f'(t1,y0+k1/2); let y1 = y0+k1/2 (a vector)."


	y1 with: y0 with: k1 set: [:_y0 :_k1 | _y0 + (_k1*0.5)].


	self dydtAt: t1 and: y1 andSet: dydt1.


	k2 with: dydt1 set: [:_dydt1 | h * _dydt1].


	"k3 = h*f'(t2,y0+k2/2); let y2 = y0+k2/2 (a vector)."


	y2 with: y0 with: k2 set: [:_y0 :_k2 | _y0 + (_k2*0.5)].


	self dydtAt: t2 and: y2 andSet: dydt2.


	k3 with: dydt2 set: [:_dydt2 | h * _dydt2].


	"k4 = h*f'(t3,y0+k3); let y3 = y0+k3 (a vector)."


	y3 with: y0 with: k3 set: [:_y0 :_k3 | _y0 + _k3].


	self dydtAt: t3 and: y3 andSet: dydt3.


	k4 with: dydt3 set: [:_dydt3 | h * _dydt3].





	"y = y0 + k1/6 + k2/3 + k3/3 + k4/6."


	solution with: y0 with: k1 with: k2 with: k3 with: k4 


		set: [:_y0 :_k1 :_k2 :_k3 :_k4 |


			_y0 + (0. 33333333 * ( 0.5 * ( _k1 + _k4) + _k2 + _k3))]





� PAGE �11�











